

21. Knobelaufgabe #9: Die Magie der Zahlen

Sie sind Kandidat*in in der Spielshow „Die Magie der Zahlen“. In der ersten Runde wird Ihnen eine Reihe von nummerierten Schachteln präsentiert, die jeweils eine für Sie nicht sichtbare ganze Zahl enthalten. Sie müssen mit möglichst wenigen Versuchen eine *magische Schachtel* finden, eine Schachtel, die ihre eigene Hausnummer enthält.

Die versteckten Zahlen sind alle unterschiedlich. Schachteln mit größeren Hausnummern enthalten größere Zahlen. Entwickeln Sie eine Strategie, um möglichst wenige Schachteln zu öffnen.

- ▶ Zum Beispiel:

1	2	3	4	5	6	7	8	9
-10	-5	0	4	7	11	27	65	99

- ▶ Es gibt nur eine magische Schachtel: #4.
- ▶ Gibt es immer eine magische Schachtel?

[Tupel](#)[Unwiderlegbare
Muster](#)[Records](#)[Varianten](#)[Motivation](#)[Abstrakte Syntax](#)[Statische Semantik](#)[Dynamische
Semantik](#)[Vertiefung](#)[Rekursive](#)[Varianten](#)[Widerlegbare
Muster](#)[Parametrisierte
Typen](#)[Polymorphie](#)[Arrays](#)

21. Motivation

Eine Person ist entweder weiblich oder männlich; eine männliche Person hat Attribute, die eine weibliche nicht hat (und vielleicht umgekehrt — aber nicht modelliert).

```
type Woman = { name : String }
```

```
type Man    = { name : String; bald : Bool }
```

Daten, die unterschiedliche Ausprägungen besitzen, können wir in Mini-F# mit sogenannten *Varianten* modellieren.

```
type Person =  
  | Female of Woman  
  | Male   of Man
```

21. Motivation

```
type Person = | Female of Woman  
              | Male   of Man
```

Eine Variantentypdefinition führt zwei verschiedene Dinge ein:

- ▶ einen Namen für den Variantentyp: *Person*,
- ▶ Namen um Elemente des Variantentyps zu konstruieren: *Female* und *Male*. Diese Bezeichner heißen auch *Datenkonstruktoren* oder kurz *Konstruktoren*.

21. Motivation

```
type Person = | Female of Woman  
              | Male   of Man
```

Umgangssprachlich lässt sich die Definition wie folgt lesen: ein Element $p : Person$ ist entweder

- ▶ von der Form *Female e* mit $e : Woman$ oder
- ▶ von der Form *Male e* mit $e : Man$.

Ein Konstruktor ähnelt einer Funktion. Der Typ nach dem Konstruktor korrespondiert zum Argumenttyp, der deklarierte Variantentyp korrespondiert zum Ergebnistyp:

- ▶ *Female* hat im Prinzip den Typ $Woman \rightarrow Person$ und
- ▶ *Male* den Typ $Man \rightarrow Person$.

☞ Im Unterschied zu einer Funktion hat ein Konstruktor aber keine Definition; er steht sozusagen für sich selbst.

21. Motivation

Beispiele:

```
Female { name = "Lisa" }  
Male { name = "Florian"; bald = false }
```

☞ Beide Ausdrücke sind vom Typ *Person*.


Ausdrücke können wie immer an Bezeichner gebunden werden.

```
let ralf      = Male { name = "Ralf";  bald = true }  
let melanie  = Female { name = "Melanie" }  
let julia    = Female { name = "Julia" }  
let andres   = Male { name = "Andres"; bald = false }
```

21. Motivation

Ein Variantentyp beschreibt Alternativen; mit Hilfe der *Fallunterscheidung* **match** können wir feststellen, welche konkrete Alternative vorliegt.

```
let dear (person : Person) : String =  
  match person with  
  | Female female → "Liebe " ^ female.name  
  | Male   male   → (if male.bald  
                    then "Lieber glatzköpfiger "  
                    else "Lieber ") ^ male.name
```

 Nach dem Schlüsselwort **match** steht der Ausdruck, der analysiert werden soll; die Zweige der Fallunterscheidung führen die verschiedenen Fälle des Variantentyps auf.

[Tupel](#)[Unwiderlegbare
Muster](#)[Records](#)[Varianten](#)**Motivation**[Abstrakte Syntax](#)[Statische Semantik](#)[Dynamische
Semantik](#)[Vertiefung](#)[Rekursive](#)[Varianten](#)[Widerlegbare
Muster](#)[Parametrisierte
Typen](#)[Polymorphie](#)[Arrays](#)

21. Variantentypen versus Baumsprachen


Variantentypen ähneln der Notation, mit der wir die abstrakte Syntax unserer Programmiersprache beschreiben.

- ▶ Baumsprachen sind Bestandteil der Sprache, mit der wir *über* die Sprache Mini-F# reden.

```
e ∈ Expr ::= false
           | true
           | num (ℕ)
           | ...
```

- ▶ Variantentypen sind Bestandteil von Mini-F#.

```
type Expr = | False
             | True
             | Num of Nat
             | ...
```

 Fachjargon: Variantentypen *internalisieren* Baumsprachen.

Tupel

Unwiderlegbare
Muster

Records

Varianten

Motivation

Abstrakte Syntax

Statische Semantik

Dynamische

Semantik

Vertiefung

Rekursive

Varianten

Widerlegbare

Muster

Parametrisierte

Typen

Polymorphie

Arrays

21. Abstrakte Syntax

Ein Variantentyp (engl. union type) wird durch eine *Definition* eingeführt.

$C \in \text{Con}$	<i>Datenkonstruktoren</i>
$d ::= \dots$	<i>Deklarationen:</i>
type $T =$ C_1 of t_1	Variantentypdefinition ($C_1 \neq C_2$)
C_2 of t_2	

☞ Der Bezeichner T wird durch die Definition neu eingeführt, ebenso die Konstruktoren C_1 und C_2 .

☞ Wie Recordtypen dürfen auch Variantentypen weder redefiniert noch lokal definiert werden.


☞ Der Bereich der Konstruktoren Con wird in Teil VI festgelegt.

Für's erste: Ein Konstruktor fängt mit einem *großen* Buchstaben an. Danach können weitere Buchstaben, kleine und große, Ziffern, und Sonderzeichen wie ein Unterstrich oder ein Apostroph folgen.

21. Abstrakte Syntax

Wir erweitern Ausdrücke um Sprachkonstrukte, die Varianten konstruieren bzw. analysieren.

$e ::= \dots$	<i>Ausdrücke:</i>
$C e$	Konstruktion
match e with $C_1 x_1 \rightarrow e_1$	Fallunterscheidung ($C_1 \neq C_2$)
$C_2 x_2 \rightarrow e_2$	

 Der Ausdruck e zwischen den Schlüsselwörtern **match** und **with** heißt *Diskriminatorausdruck*; $C_1 x_1 \rightarrow e_1$ und $C_2 x_2 \rightarrow e_2$ sind *Zweige* der Fallunterscheidung.

21. Statische Semantik

Die folgenden Typregeln setzen voraus, dass der Variantentyp

type $T = | C_1 \text{ of } t_1 | C_2 \text{ of } t_2$

bekannt ist.

Typregeln:

$$\frac{\Sigma \vdash e : t_i}{\Sigma \vdash C_i e : T}$$

$$\frac{\Sigma \vdash e : T \quad \Sigma, \{x_1 \mapsto t_1\} \vdash e_1 : t \quad \Sigma, \{x_2 \mapsto t_2\} \vdash e_2 : t}{\Sigma \vdash (\mathbf{match} \ e \ \mathbf{with} \ | C_1 \ x_1 \rightarrow e_1 \ | C_2 \ x_2 \rightarrow e_2) : t}$$

☞ Alle Zweige der Fallunterscheidung müssen den gleichen Typ besitzen; dieser ist auch der Typ des gesamten Ausdrucks.

Tupel

Unwiderlegbare
Muster

Records

Varianten

Motivation

Abstrakte Syntax

Statische Semantik

Dynamische

Semantik

Vertiefung

Rekursive

Varianten

Widerlegbare

Muster

Parametrisierte

Typen

Polymorphie

Arrays

21. Dynamische Semantik

Konstruktoren konstruieren Werte, entsprechend müssen wir den Bereich der Werte erweitern.

$$\nu ::= \dots \\ | C \nu$$

Werte:
Konstruktion \setminus Injektion in einen Variantentyp

Auswertungsregeln:

$$\frac{\delta \vdash e \Downarrow \nu}{\delta \vdash C e \Downarrow C \nu}$$

$$\frac{\delta \vdash e \Downarrow C_i \nu_i \quad \delta, \{x_i \mapsto \nu_i\} \vdash e_i \Downarrow \nu}{\delta \vdash (\mathbf{match} \ e \ \mathbf{with} \ | C_1 \ x_1 \rightarrow e_1 \ | C_2 \ x_2 \rightarrow e_2) \Downarrow \nu}$$

21. Varianten mit n Alternativen

Alle Konstrukte verallgemeinern sich in natürlicher Weise auf Varianten mit n Alternativen.

- ▶ $n = 0$:
 - ▶ keine Alternative: leerer Typ;

```
type Empty = |
```

- ▶ Fallunterscheidung ohne Zweige: **match** *e* **with**;
- ▶ die leere Fallunterscheidung signalisiert *toten Code*:

```
you-cannot-call-me (x : Empty) : Nat = match x with
```

 F# kennt keine leeren Varianten.

21. Varianten mit n Alternativen

- ▶ $n = 1$:
 - ▶ eine Alternative:

```
type Price    = | Cent of Nat
type Postcode = | Code of Nat
```

- ▶ Fallunterscheidung hat genau einen Zweig:

```
let double (price : Price) : Price =
  match price with | Price n → Price (2 * n)
```

- ▶ 1-Varianten sehr nützlich: sind *Price* und *Postcode* wie oben definiert, so stellt die statische Semantik sicher, dass wir in einem Programm nicht aus Versehen Preise und Postleitzahlen addieren.
- ▶ Auch lässt sich ein Preis p nicht mit $2 * p$ verdoppeln. Zu diesem Zweck muss *double* verwendet werden.
- ▶ Der Gewinn an Sicherheit wird mit einem Verlust an Bequemlichkeit erkaufte.
- ▶ $n = 3$:
 - ▶ drei Alternativen und Fallunterscheidungen;
- ▶ ...

[Tupel](#)[Unwiderlegbare Muster](#)[Records](#)[Varianten](#)[Motivation](#)[Abstrakte Syntax](#)[Statische Semantik](#)[Dynamische Semantik](#)[Vertiefung](#)[Rekursive Varianten](#)[Widerlegbare Muster](#)[Parametrisierte Typen](#)[Polymorphie](#)[Arrays](#)

21. Vertiefung

Der Typ *Bool* kann durch einen Variantentyp implementiert werden.

```
type Bool = | False of Unit | True of Unit
```

- ▶ *false* wird durch *False ()* repräsentiert,
- ▶ *true* wird durch *True ()* repräsentiert,
- ▶ **if** e_1 **then** e_2 **else** e_3 wird durch die Fallunterscheidung **match** e_1 **with** | *False ()* $\rightarrow e_3$ | *True ()* $\rightarrow e_2$ realisiert.

„Nullstellige“ Konstruktoren wie *False* oder *True* sind relativ häufig. Aus diesem Grund erlauben wir, das Dummyargument auch wegzulassen.

21. Fallstudie: Ganze Zahlen — da capo

Idee: eine ganze Zahl wird durch das Vorzeichen und den Betrag repräsentiert.

```
type Int = | Neg of Nat | Pos of Nat
```

☞ *Neg* n bezeichnet die Zahl $-n$; *Pos* n entsprechend die Zahl $+n$.

☞ *Invariante*: 0 wird durch *Pos* 0 repräsentiert; für *Neg* n gilt stets $n > 0$.

Cleverer Konstruktor:

```
let neg (n : Nat) : Int =  
  if n = 0 then Pos 0 else Neg n
```

☞ *neg* etabliert die Invariante.

21. Ganze Zahlen — Klassifikation

```
let is-negative (i : Int) : Bool =
```

```
  match i with
```

```
    | Neg n → true
```

```
    | Pos n → false
```

```
let is-zero (i : Int) : Bool =
```

```
  match i with
```

```
    | Neg n → false
```

```
    | Pos n → n = 0
```

```
let is-positive (i : Int) : Bool =
```

```
  match i with
```

```
    | Neg n → false
```

```
    | Pos n → true
```


21. Ganze Zahlen — arithmetische Operationen

```
let negate (i : Int) : Int =
```

```
  match i with
```

```
    | Neg n → Pos n
```

```
    | Pos n → neg n
```

```
let add (i : Int, j : Int) : Int =
```

```
  match (i, j) with
```

```
    | (Neg m, Neg n) → Neg (m + n)
```

```
    | (Neg m, Pos n) → if m ≤ n then Pos (n ÷ m)
```

```
                        else Neg (m ÷ n)
```

```
    | (Pos m, Neg n) → if m < n then Neg (n ÷ m)
```

```
                        else Pos (m ÷ n)
```

```
    | (Pos m, Pos n) → Pos (m + n)
```

```
let sub (i : Int, j : Int) : Int =
```

```
  add (i, negate j)
```

21. Ganze Zahlen — arithmetische Operationen

```
let mul (i : Int, j : Int) : Int =  
  match (i, j) with  
  | (Neg m, Neg n) → Pos (m * n)  
  | (Neg m, Pos n) → neg (m * n)  
  | (Pos m, Neg n) → neg (m * n)  
  | (Pos m, Pos n) → Pos (m * n)
```

☞ *mul* implementiert die Vorzeichenregel: $- * - = +$ etc.

☞ *div* und *mod* zur Übung.

21. Ganze Zahlen — Konversion und Betrag

```
let int (n : Nat) : Int = Pos n
```

```
let abs (i : Int) : Nat =
```

```
  match i with
```

```
    | Neg n → n
```

```
    | Pos n → n
```

21. Ganze Zahlen — Vergleichsoperationen

```
let less (i : Int, j : Int) : Bool =  
  match (i, j) with  
    | (Neg m, Neg n) → m > n  
    | (Neg m, Pos n) → true  
    | (Pos m, Neg n) → false  
    | (Pos m, Pos n) → m < n
```

```
let less-equal (i : Int, j : Int) : Bool =  
  match (i, j) with  
    | (Neg m, Neg n) → m ≥ n  
    | (Neg m, Pos n) → true  
    | (Pos m, Neg n) → false  
    | (Pos m, Pos n) → m ≤ n
```

```
let equal (i : Int, j : Int) : Bool =  
  match (i, j) with  
    | (Neg m, Neg n) → m = n  
    | (Neg m, Pos n) → false  
    | (Pos m, Neg n) → false  
    | (Pos m, Pos n) → m = n
```

21. Ganze Zahlen — Vergleichsoperationen

```
let not-equal (i : Int, j : Int) : Bool =
```

```
  match (i, j) with
```

```
    | (Neg m, Neg n) → m <> n
```

```
    | (Neg m, Pos n) → true
```

```
    | (Pos m, Neg n) → true
```

```
    | (Pos m, Pos n) → m <> n
```

```
let greater-equal (i : Int, j : Int) : Bool =
```

```
  match (i, j) with
```

```
    | (Neg m, Neg n) → m ≤ n
```

```
    | (Neg m, Pos n) → false
```

```
    | (Pos m, Neg n) → true
```

```
    | (Pos m, Pos n) → m ≥ n
```

```
let greater (i : Int, j : Int) : Bool =
```

```
  match (i, j) with
```

```
    | (Neg m, Neg n) → m < n
```

```
    | (Neg m, Pos n) → false
```

```
    | (Pos m, Neg n) → true
```

```
    | (Pos m, Pos n) → m > n
```

21. Ganze Zahlen — Demo

Mini) *int* 4711

Pos 4711

Mini) *add* (*negate* (*int* 4711), *int* 815)

Neg 3896

Mini) *add* (*int* 4711, *negate* (*int* 4711))

Pos 0

Mini) *is-zero* (*add* (*int* 4711, *negate* (*int* 4711)))

true

Mini) *mul* (*negate* (*int* 2), *negate* (*int* 3))

Pos 6

Mini) *abs* (*mul* (*negate* (*int* 2), *int* 3))

6

Mini) *div* (*negate* (*int* 4), *int* 3)

Neg 2

Mini) *mod* (*negate* (*int* 4), *int* 3)

Pos 2

☞ Für $b \neq 0$ gilt weiterhin $a = (a \div b) * b + (a \% b)$. Aber: die Eigenschaft legt \div und $\%$ nicht länger eindeutig fest, siehe Übung.

21. Summen und Produkte

Ist t ein endlicher Typ, so bezeichnet $|t|$ die Anzahl der Elemente von t , die *Kardinalität* von t .

1. Der Paartyp $t_1 * t_2$ korrespondiert zu einem *Produkt*, da

$$|t_1 * t_2| = |t_1| * |t_2|$$

2. Der Typ *Unit* korrespondiert zu der 1, da


$$|Unit| = 1$$

3. Der Variantentyp T mit **type** $T = \mid C_1 \text{ of } t_1 \mid C_2 \text{ of } t_2$ korrespondiert zu einer *Summe*, da

$$|T| = |t_1| + |t_2|$$

4. Der Variantentyp *Empty* mit **type** $Empty = \mid$ korrespondiert zu der 0, da

$$|Empty| = 0$$

 Und Funktionen?

Tupel

Unwiderlegbare
Muster

Records

Varianten

Motivation

Abstrakte Syntax

Statische Semantik

Dynamische

Semantik

Vertiefung

Rekursive

Varianten

Widerlegbare
MusterParametrisierte
Typen

Polymorphie

Arrays

21. Summen und Produkte — Eigenschaften

Für Paar- und Variantentypen gelten ähnliche Gesetze wie für die natürlichen Zahlen. Zum Beispiel:

$$0 + t \cong t \cong t + 0$$

$$0 \times t \cong 0 \cong t \times 0$$

$$1 \times t \cong t \cong t \times 1$$

$$t_1 \times (t_2 + t_3) \cong (t_1 \times t_2) + (t_1 \times t_3)$$

$$(t_1 + t_2) \times t_3 \cong (t_1 \times t_3) + (t_2 \times t_3)$$

☞ Im Unterschied zu den natürlichen Zahlen sind die beiden oder die drei Seiten nicht gleich, sondern nur *isomorph*: $t_1 \cong t_2$ bedeutet, dass sich jedes Element aus t_1 eindeutig einem Element aus t_2 zuordnen lässt.

☞ Und Funktionen?

[Tupel](#)[Unwiderlegbare
Muster](#)[Records](#)[Varianten](#)[Motivation](#)[Abstrakte Syntax](#)[Statische Semantik](#)[Dynamische](#)[Semantik](#)[Vertiefung](#)[Rekursive](#)[Varianten](#)[Widerlegbare
Muster](#)[Parametrisierte
Typen](#)[Polymorphie](#)[Arrays](#)

21. Rechnen mit Typen

Der Typ *Person* in arithmetischer Notation:

$$Person \cong String + (String \times Bool)$$

Jetzt können wir rechnen:

$$\begin{aligned} & String + (String \times Bool) \\ \cong & \{ 1 \text{ ist das neutrale Element von } ' \times ' \} \\ & (String \times 1) + (String \times Bool) \\ \cong & \{ \text{Distributivgesetz} \} \\ & String \times (1 + Bool) \end{aligned}$$

☞ Der Typ $String \times (1 + Bool)$ ist eine alternative Implementierung von *Person*.

☞ Für das Rechnen mit Typen (!) ist die arithmetische Notation sehr bequem — es ist aber nur eine Notation!

Tupel

Unwiderlegbare
Muster

Records

Varianten

Motivation

Abstrakte Syntax

Statische Semantik

Dynamische
Semantik

Vertiefung

Rekursive
VariantenWiderlegbare
MusterParametrisierte
Typen

Polymorphie

Arrays

21. Rechnen mit Typen

Übersetzen wir den Typ $String \times (1 + Bool)$ in Mini-F# Notation, so erhalten wir

```
type Person' = { name : String; gender : Gender }
```

```
type Gender = Female' | Male' of { bald : Bool }
```

☞ Das Geschlecht umfasst die trennenden Merkmale, die gemeinsamen sind in *Person'* zusammengefasst.

22. Motivation

- ▶ Mit den Konstrukten, die wir bisher eingeführt haben, können wir nur eine beschränkte Anzahl von Daten zusammenfassen:
 - ▶ ein 7-Tupel aggregiert 7 Daten,
 - ▶ ein 128-Tupel 128 Daten.
- ☞ Beide Typen sind ungeeignet um 6, 8, 127 oder 129 Daten aufzunehmen.
- ▶ Zum Zeitpunkt des Programmierens kennt man häufig die genaue Anzahl von Daten nicht:
 - ▶ Wieviele Personen immatrikulieren sich im WS 2019/2020?
 - ▶ Wieviele Unternehmen sind an der Börse notiert?
 - ▶ Wieviele Mitarbeiter*innen hat eine Abteilung?
 - ▶ usw.
- ▶ Wie können wir eine beliebige Anzahl von Daten aggregieren?

22. Motivation

Konkret: wie können wir eine Folge von natürlichen Zahlen repräsentieren?

Ein erster Versuch (in arithmetischer Notation):

$$\begin{aligned} \mathit{Nats} &\cong 1 \\ &+ \mathit{Nat} \\ &+ \mathit{Nat} \times \mathit{Nat} \\ &+ \mathit{Nat} \times \mathit{Nat} \times \mathit{Nat} \\ &+ \dots \end{aligned}$$

☞ Eine Folge von natürlichen Zahlen ist entweder die leere Folge (ein 0-Tupel), oder eine einelementige Folge (ein „1-Tupel“), oder eine zweielementige Folge (ein 2-Tupel) usw.

22. Motivation

Beobachtung: alle Alternativen bis auf die erste haben eine *Nat* Komponente. Der gemeinsame Faktor kann „herausgezogen“ werden.

$$\begin{aligned} \mathit{Nats} &\cong 1 \\ &+ \mathit{Nat} \times (1 \\ &\quad + \mathit{Nat} \\ &\quad + \mathit{Nat} \times \mathit{Nat} \\ &\quad + \dots) \end{aligned}$$


22. Motivation

Der Typausdruck in Klammern entspricht der ursprünglichen Definition von *Nats*.

$$\mathit{Nats} \cong 1 + \mathit{Nat} \times \mathit{Nats}$$

Erlauben wir bei der Definition eines Typs den Rückgriff auf den definierten Typ selbst (!), erhalten wir (in Mini-F# Notation):

```
type Nats = | Nil | Cons of Nat * Nats
```

 *Zur Erinnerung:* Greift man bei der Definition auf das definierte Objekt selbst zurück, spricht man von einer *rekursiven* Definition.

22. Listen

```
type Nats = | Nil | Cons of Nat * Nats
```

☞ Eine Folge von natürlichen Zahlen ist entweder die leere Folge *Nil* oder eine mindestens einelementige Folge *Cons* (n, ns) bestehend aus einer natürlichen Zahl n und einer Folge von natürlichen Zahlen ns .

-
- ▶ *Nil* ist eine Verkürzung des lateinischen Wortes *nihil* für „nichts“.
 - ▶ *Cons* kürzt das englische Wort *construct* ab.
 - ▶ Statt von einer Folge von natürlichen Zahlen spricht man auch kurz von einer *Liste*:
 - ▶ n ist das *Kopfelement* der Liste *Cons* (n, ns),
 - ▶ ns ist die *Restliste* der Liste *Cons* (n, ns).
 - ▶ Bei Listen ist wie bei Tupeln die Reihenfolge der Elemente signifikant.
 - ▶ Listen sind die erste und einfachste *Datenstruktur*, die uns begegnet. Eine Datenstruktur verwaltet Daten und unterstützt Zugriff und Manipulation dieser Daten.

22. Listen — Programmierung

Wie gehen wir mit einem rekursiven Variantentyp um? Wir konstruieren und analysieren rekursive Varianten mit Hilfe rekursiver Funktionen!

Beispiel: Sortieren von Listen. Der Variantentyp gibt das folgende Skelett für *sort* vor.

```
let rec sort (nats : Nats) : Nats =  
  match nats with  
  | Nil           → ...  
  | Cons (n, ns) → ...
```

☞ Wie füllen wir den *Cons* Zweig mit Leben?

22. Sortieren

Getreu dem Motto „rekursive Funktionen für rekursive Typen“ erlauben wir, *sort* auf die Restliste *ns* anzuwenden.

```
let rec sort (nats : Nats) : Nats =  
  match nats with  
  | Nil           → ...  
  | Cons (n, ns) → ... sort ns ...
```

22. Sortieren

- ▶ *Rekursionsbasis*: die leere Liste ist bereits geordnet.

```
let rec sort (nats : Nats) : Nats =
  match nats with
  | Nil          → Nil
  | Cons (n, ns) → ... sort ns ...
```

- ▶ *Rekursionsschritt*: *sort ns* ist eine geordnete Liste; wir müssen das Element *n* an die richtige Stelle einordnen. Wir geben dieser Teilaufgabe einen Namen: *insert*.

```
let rec sort (nats : Nats) : Nats =
  match nats with
  | Nil          → Nil
  | Cons (n, ns) → insert (n, sort ns)
```

☞ Wenn sich ein Problem nicht mit dem bisherigen Vokabular lösen lässt, müssen wir das Vokabular erweitern. An dieser Stelle ist Kreativität gefragt!

Tupel

Unwiderlegbare
Muster

Records

Varianten

Rekursive
VariantenMotivation
VertiefungWiderlegbare
MusterParametrisierte
Typen

Polymorphie

Arrays

22. Einfügen in eine geordnete Liste

Die Definition von *insert* gehen wir auf die gleiche Art und Weise an.

```
let rec insert (nat : Nat, nats : Nats) : Nats =  
  match nats with  
  | Nil           → ...  
  | Cons (n, ns) → ... insert (nat, ns) ...
```

Vorbedingung: *nats* ist bereits geordnet. Das Typsystem stellt diese Eigenschaft nicht sicher, darum müssen wir uns kümmern.

[Tupel](#)[Unwiderlegbare
Muster](#)[Records](#)[Varianten](#)[Rekursive
Varianten](#)[Motivation
Vertiefung](#)[Widerlegbare
Muster](#)[Parametrisierte
Typen](#)[Polymorphie](#)[Arrays](#)

22. Einfügen in eine geordnete Liste


- ▶ *Rekursionsbasis*: Ist die Liste leer, so geben wir die einelementige Liste *Cons (nat, Nil)* zurück.

```
let rec insert (nat : Nat, nats : Nats) : Nats =
  match nats with
  | Nil          → Cons (nat, Nil)
  | Cons (n, ns) → ...
```

- ▶ *Rekursionsschritt*: gilt $nat \leq n$, so stellen wir *nat* an den Anfang der Liste; anderenfalls wird *nat* in die Restliste *ns* eingefügt.

```
let rec insert (nat : Nat, nats : Nats) : Nats =
  match nats with
  | Nil          → Cons (nat, Nil)
  | Cons (n, ns) → if nat ≤ n
                    then Cons (nat, nats)
                    else Cons (n, insert (nat, ns))
```

22. Sortieren durch Einfügen



22. Sortieren — Verallgemeinerung

Die Funktion *sort* sortiert eine Liste aufsteigend. Was machen wir, wenn wir die Liste absteigend ordnen wollen?

- ▶ Programmcode duplizieren und ' \leq ' systematisch durch ' \geq ' ersetzen. Unökonomisch!
- ▶ Aufsteigend sortieren und das Ergebnis umdrehen, siehe Skript.
- ▶ Wir verallgemeinern die Aufgabenstellung und abstrahieren von einer speziellen Ordnungsrelation.

```
sort-by (less-equal : Nat * Nat → Bool) : Nats → Nats
```

22. Sortieren — Verallgemeinerung

Sortieren durch Einfügen

```
let sort-by (less-equal : Nat * Nat → Bool) : Nats → Nats =  
  let rec insert (nat : Nat, nats : Nats) : Nats =  
    match nats with  
    | Nil          → Cons (nat, Nil)  
    | Cons (n, ns) → if less-equal (nat, n)  
                     then Cons (nat, nats)  
                     else Cons (n, insert (nat, ns))  
  
  let rec sort (nats : Nats) : Nats =  
    match nats with  
    | Nil          → Nil  
    | Cons (n, ns) → insert (n, sort ns)  
  
in sort
```

 die Definition von *sort-by* verwendet Layout (Abseitsregel).

[Tupel](#)[Unwiderlegbare
Muster](#)[Records](#)[Varianten](#)[Rekursive
Varianten](#)[Motivation
Vertiefung](#)[Widerlegbare
Muster](#)[Parametrisierte
Typen](#)[Polymorphie](#)[Arrays](#)

22. Sortieren — Verallgemeinerung

Die ursprünglichen Sortierfunktionen sind jetzt hausbackene Spezialfälle:

```
let increasing-sort = sort-by (fun (m, n) → m ≤ n)
```

```
let decreasing-sort = sort-by (fun (m, n) → m ≥ n)
```

☞ Die Funktion *sort-by* ist ein weiteres Beispiel für eine Funktion höherer Ordnung: *sort-by* nimmt eine Funktion als Argument (*less-equal*) und gibt eine Funktion (*sort*) als Ergebnis zurück.

22. Struktur Entwurfsmuster für Listen

Haben wir die Aufgabe eine Funktion $f : Nats \rightarrow t$ zu erstellen, dann sieht ein erster Entwurf folgendermaßen aus.

let rec $f (nats : Nats) : t =$	<i>Struktur Entwurfsmuster</i>
match $nats$ with	
<i>Nil</i> $\rightarrow \dots$	<i>Rekursionsbasis</i>
<i>Cons</i> $(n, ns) \rightarrow \dots n \dots f ns \dots$	<i>Rekursionsschritt</i>

Die Ellipsen müssen mit Leben gefüllt werden:

- ▶ *Rekursionsbasis*: ein Ausdruck des Typs t .
- ▶ *Rekursionsschritt*: ein Ausdruck, der die Teillösung $f ns$ vom Typ t zu einer Gesamtlösung vom Typ t erweitert.

22. Summe einer Liste von Zahlen

Aufgabe: $sum\ nats$ soll die Elemente der Liste $nats$ aufaddieren (die Verallgemeinerung von $+$ auf Listen).

```
let rec sum (nats : Nats) : Nat =  
  match nats with  
  | Nil           → ...  
  | Cons (n, ns) → ... sum ns ...
```

22. Summe einer Liste von Zahlen

- ▶ *Rekursionsbasis*: $\text{sum Nil} = 0$. Warum?

```
let rec sum (nats : Nats) : Nat =  
  match nats with  
  | Nil           → 0  
  | Cons (n, ns) → ... sum ns ...
```

- ▶ *Rekursionsschritt*: Wir addieren n zur Summe der Restliste.

```
let rec sum (nats : Nats) : Nat =  
  match nats with  
  | Nil           → 0  
  | Cons (n, ns) → n + sum ns
```

[Tupel](#)[Unwiderlegbare
Muster](#)[Records](#)[Varianten](#)[Rekursive
Varianten](#)[Motivation
Vertiefung](#)[Widerlegbare
Muster](#)[Parametrisierte
Typen](#)[Polymorphie](#)[Arrays](#)

22. Produkt einer Liste von Zahlen

Aufgabe: *product nats* soll die Elemente der Liste *nats* miteinander multiplizieren (die Verallgemeinerung von $*$ auf Listen).

```
let rec product (nats : Nats) : Nat =  
  match nats with  
  | Nil           → ...  
  | Cons (n, ns) → ... product ns ...
```

22. Produkt einer Liste von Zahlen

- ▶ *Rekursionsbasis: product Nil = 1. Warum?*

```
let rec product (nats : Nats) : Nat =  
  match nats with  
  | Nil          → 1  
  | Cons (n, ns) → ... product ns ...
```

- ▶ *Rekursionsschritt: Wir multiplizieren n mit dem Produkt der Restliste.*

```
let rec product (nats : Nats) : Nat =  
  match nats with  
  | Nil          → 1  
  | Cons (n, ns) → n * product ns
```

22. Konstruktion von Listen

Die bisherigen Funktionen verarbeiten Listen; wie können wir Listen erzeugen?

Beispiel: `between (l, u)` erzeugt die Liste aller Elemente in dem gegebenen Intervall.

Wir wenden das Peano Entwurfsmuster auf die Intervallgröße an.

```
let rec between (l : Nat, u : Nat) : Nats =  
  if l > u then Nil  
    else Cons (l, between (l + 1, u))
```


☞ Im Basisfall geben wir die leere Liste zurück; im Rekursionsfall setzen wir die linke Intervallgrenze vor die rekursiv erzeugte Liste.

[Tupel](#)[Unwiderlegbare
Muster](#)[Records](#)[Varianten](#)[Rekursive
Varianten](#)[Motivation
Vertiefung](#)[Widerlegbare
Muster](#)[Parametrisierte
Typen](#)[Polymorphie](#)[Arrays](#)

22. Alte Funktionen neu

Wir haben unser Vokabular beträchtlich erweitert. Mit den neuen Vokabeln können wir zum Beispiel *factorial* kürzer definieren.

```
let factorial (n : Nat) : Nat = product (between (1, n))
```

 Wie lässt sich *power* auf *product* zurückführen?